
APLICAŢII 
 

1. Calcularea rutinelor de întârziere 
 
a) Exemplu de rutină de întârziere: 
 
CALL DELAY1   ;apelez rutina de întârziere  (în 2µs)  
 
DELAY1:    
 MOV R0,#46  ;încarc registrul R0 cu constanta 46 zecimal   (în 2µs)  
DELAY10: 
 DJNZ R0,DELAY10 ;decrementez R0. Dacă R0 nu este egal cu zero, sari la DELAY1 (în 46*2µs)    
 RET   ;întoarcere din apelare (în 2µs)  
 
Să facem un simplu calcul matematic: 
 2µs  + 2µs + 46*2µs + 2µs = 98µs   
 
 
b) Exemplu de rutină de întârziere: 
 
CALL DELAY2   ;apelez rutina de întârziere  (în 2µs)  
 
DELAY2:    
 MOV R0,#8   ;încarc registrul R0 cu, constanta 8 zecimal   (în 2µs) 

MOV R1,#196  ;încarc registrul R1 cu, constanta 196 zecimal   (în 2µs)  
DELAY20: 
 DJNZ R1,DELAY20 ; decrementez R1. Dacă R1 nu este egal cu zero, sari la DELAY20  
              DJNZ R0,DELAY20 ;decrementez R0. Dacă R0 nu este egal cu zero, sari la DELAY20 
 RET   ;întoarcere din apelare (în 2µs)  
 
Să facem un simplu calcul matematic: 
 2µs  + 2µs + 2µs + 196*2µs + 256*2µs*(8-1) + 8*2µs + 2µs = 4000µs = 4ms 
 
 
c) Exemplu de rutină de întârziere: 

Să presupunem că ne dorim calcularea unei rutine de întârziere de 100µs. Din primul moment, îmi voi scădea cele  2µs – 
corespunzător instr. CALL şi  2µs – corespunzător instr. RET. 

100µs - 2µs - 2µs = 96µs 
Timpul rămas, îl împart la 2 (instr. DJNZ este pe 2µs). 
 96µs / 2 = 48µs 
În mod normal, ar trebui să încarc registrul ales (de exp. R0), cu valoarea 48 (am lucrat NUMAI în zecimal); dar să nu uităm că 
şi timpul de încărcare al Registrului ales, durează şi el 2µs (exp. MOV R0,#xxx) → atunci voi face în felul următor: (96µs-
2µs)/2=47. Abia această valoare o încarc în Registrul ales.  
Intuitiv, puteam din start să scad cele 2µs necesare execuţiei (exp. MOV R0,#xxx), dar acest lucru este valabil doar la Rutine de 
întârziere mici (ce folosesc doar un singur Registru de lucru. exp. R0, R1, R2,..., un octet din memoria de date (exp. 40H), etc.).  
 
CALL DELAY1   ;apelez rutina de întârziere  (în 2µs)  
 
DELAY1:    
 MOV R0,#47  ;încarc registrul R0 cu constanta 47 zecimal   (în 2µs)  
DELAY10: 
 DJNZ R0,DELAY10 ;decrementez R0. Dacă R0 nu este egal cu zero, sari la DELAY1 (în 47*2µs)    
 RET   ;întoarcere din apelare (în 2µs) 

d) Exemplu de rutină de întârziere: 
 Să presupunem că ne dorim calcularea unei rutine de întârziere de 10ms. Din primul moment, îmi voi scădea cele  2µs – 
corespunzător instr. CALL şi  2µs – corespunzător instr. RET. 
 10ms - 2µs - 2µs = 9.996µs 
Dacă împart 9.996 la 2 rezultă 4.998, valoare ce nu pot să o încarc într-un octet (val. max. este 255). În această situaţie mai am 
nevoie de un alt registru. Atunci: 
 9.996 / 256 = 39,03   
Această nouă valoare este împărţită la 2 (deoarece instr. DJNZ este executată în 2µs). 
 39,03  / 2 = 19,5 
Deoarece 19,5 nu este nr. întreg ; nu poate fi memorat intr-un registru. Voi trunchia această valoare, obţin 19. 
Reiau calculul cu noua valoare: 

19*2(µs) =38 – se înmulţeşte cu 2 deoarece DJNZ se execută în 2µs. 
38*256 = 9.728(µs) – această valoare se obţine cu cel de-al doilea registru.  

Cu ce valoare încarc primul registru ? 
9.996 – 9.728 = 268 

Din această valoare scad timpul necesar încărcării celor două registre: 
268 - 2 - 2 = 264 

Tot din acest timp, mai scad timpul de execuţie al instr. DJNZ de la celălalt registru: 
264 – (19 + 1) *2µs = 224µs – se înmulţeşte cu 2 deoarece DJNZ se execută în 2µs. 

 IMPORTANT: observaţi apariţia în paranteză a cifrei 1!  
Abia în acest moment obţin valoarea de încărcare pt. primul registru: 

224 / 2µs = 112 
  
CALL DELAY2   ;apelez rutina de întârziere  (în 2µs)  
 
DELAY2:    
 MOV R2,#20   ;încarc registrul R0 cu, constanta 20 zecimal   (în 2µs) 

MOV R1,#112  ;încarc registrul R1 cu, constanta 112 zecimal   (în 2µs)  
DELAY20: 
 DJNZ R1,DELAY20 ; decrementez R1. Dacă R1 nu este egal cu zero, sari la DELAY20  
              DJNZ R2,DELAY20 ;decrementez R0. Dacă R0 nu este egal cu zero, sari la DELAY20 
 RET 
 
Observaţi că Reg. R2 este încărcat cu valoarea 20 , şi nu cu 19, aşa cum ne-am fi aşteptat; apare 1 în plus deoarece mai întâi este 
decrementat R1, şi apoi de 19 ori se execută: (19*2 µs)*256 
 
 
d) Exemplu de rutină de întârziere: 

Să presupunem că ne dorim calcularea unei rutine de întârziere de 20ms. Din primul moment, îmi voi scădea cele  2µs – 
corespunzător instr. CALL şi  2µs – corespunzător instr. RET. 

20ms - 2µs - 2µs = 19.996µs 
19,996 / 256 = 78,109 
78,109 / 2(µs) = 39,054 
Trunchiez 39,054, rezultă: 39. 
Recalculez cu noua valoare: 
(39*2µs)*256 = 19.968 µs 
Cu cât încarc primul registru? 
19,996 µs – 19.968 µs = 28 µs 
28 µs - 2 µs -2 µs = 24 µs  (am scăzut timpul de încărcare al celor două registre) 
24µs – (39*2µs) = -54 µs 
Nu este bine.  
Am să scad o unitate de la registrul al doilea: 39 – 1 = 38. Recalculez: 
(38*2µs)*256 = 19.456 µs 
Cu cât încarc primul registru? 
19,996 µs – 19.456 µs = 540 µs 
540 µs - 2 µs -2 µs = 536 µs  (am scăzut timpul de încărcare al celor două registre) 
536µs – (38 + 1)*2µs = 458µs 
Obţin valoarea de încărcare pentru primul registru: 
458 µs / 2 µs = 229 

CALL DELAY2   ;apelez rutina de întârziere  (în 2µs)  
 
DELAY2:    
 MOV R2,#39   ;încarc registrul R0 cu, constanta 39 zecimal   (în 2µs) 

MOV R1,#229  ;încarc registrul R1 cu, constanta 229 zecimal   (în 2µs)  
DELAY20: 
 DJNZ R1,DELAY20 ; decrementez R1. Dacă R1 nu este egal cu zero, sari la DELAY20  
              DJNZ R2,DELAY20 ;decrementez R0. Dacă R0 nu este egal cu zero, sari la DELAY20 
 RET 
 
 

2. Program:  Generarea unui semnal sonor  (cu frecvenţa de 1KHz) 
 
 
 
 
 
 
 
 
 
 
 
 
 
 

 
Să presupunem că vrem generarea unui semnal sonor cu frecvenţa de 1KHz. Va trebui mai întâi să 
calculăm perioada semnalului. 

msunde
KHzfrecvanta

Tperioada 1sec10
1

11 3 ==== −  

Vom împărţi această Perioadă în doua segmente de timp egale, astfel încât, la un bit dintr-un port să am 
nivel de tensiune 1 logic = 0,5ms şi nivel de tensiune 0 logic = 0,5ms; apoi iar să am nivel de tensiune 1 
logic = 0,5ms şi iar nivel de tensiune 0 logic = 0,5ms, etc. 
 
Programul: 

name ton    ;nume program 
ORG 0000H   ;INCEPE PROGRAMUL DE LA ADRESA 0000H 
BUZZ EQU P3.0   ;ETCHETEAZA PINUL DE PORT P3.0 
START: 
 MOV P3,#00H  ;STINGE TOATE LED-urile CONECTATE LA PORTUL P3 - daca am asa ceva... 
 SETB BUZZ  ;PUNE IN 1 LOGIC BITUL P3.0 
 CALL DELAY  ;APELEAZA O INTARZIERE DE 500 MICROSECUNDE 
 CLR BUZZ  ;PUNE IN 0 LOGIC BITUL P3.0 
 CALL DELAY  ;APELEAZA O INTARZIERE DE 500 MICROSECUNDE 
 JMP START  ;SARI LA START SI REPETA 
  
DELAY:    ;RUTINA DE INTARZIERE DE 500 uS 
 MOV R0,#246  ;INCARCA REGISTRUL R0 CU CONSTANTA DIRECTA 246 ZECIMAL 
DELAY1: 
 DJNZ R0,DELAY1 ;DECREMENTEAZA R0 PANA CAND ACESTA ESTE EGAL CU ZERO 
 NOP   ;PIERDE TIMPUL 1uS 
 NOP   ;PIERDE TIMPUL 1uS 
 RET   ;INTOARCERE DIN SUBRUTINA DE INTARZIERE 
 END   ;SFARSIT DE PROGRAM  
 
 

Nivel logic
de tensiune

1 logic

0 logic
perioadaT

2
perioadaT

2
perioadaT

timp

Program:  Generarea unui semnal sonor cu două tonuri 
 

 
 
 
 
 
 
 
 
 
 
 
 
 
 

 
Să presupun că vreau ca cele două frecvenţe să fie de F1 = 400Hz şi F2 = 2KHz, ceea ce 

corespunde unor perioade de msTperioada 5,21 = şi msTperioada 5,02 = . Perioada unui semnal este formată 
din doua semialternanţe ce corespunde celor doua nivele logice de zero şi de 1 logic. Pentru generarea 
semnalului de frecventa 400Hz, va trebui ca să „stau” pe 1 logic 2,5ms/2 şi pe zero logic 2,5ms/2. 
Repet această perioadă (ce îmi dă frecvenţa semnalului meu de 400Hz) de cate ori vreau eu, iar dacă 
vreau să am semnal de această frecvenţă timp de 1 secundă, va trebui să repet această perioadă de: 
1000ms / 2,5ms = 400 ori. La fel se întâmplă şi cu semnalul de frecvenţă 2KHz. Repet perioada de 
0,5ms , într-o secundă de: 1000ms / 0,5ms = 2000 ori.   
 
 
Programul: 
 

name biton 
ORG 0000H 
BUZZ EQU P3.0 
START: 
 MOV R0,#144 ;STABILESC PRIN R0 SI R1 REPETAREA DE 400 ORI A LUI F1 
 MOV R1,#2 ;144 + 256*1 = 400 
 MOV R2,#208 ;STABILESC PRIN R2 SI R3 REPETAREA DE 2000 ORI A LUI F2 
 MOV R3,#8 ;208 + 256*7 = 2000 
 
F1:   SETB BUZZ ;APLIC 1 LOGIC LA BUZZ 
 CALL DELAY1 ;ASTEPT 1,25 MILISECUNDE 
 CLR BUZZ ;APLIC 0 LOGIC LA BUZZ 
 CALL DELAY1 ;ASTEPT 1,25 MILISECUNDE 
 DJNZ R0,F1 ;REPET CELE DOUA SEMIPERIOADE (1,25ms + 1,25ms =2,5ms) CE-MI DA O                                                                 
   ;FRECVENTA  
 DJNZ R1,F1 ;A SEMNALULUI AUDIO DE 400 Hz,  DE 400 ORI , ADICA 400*2,5ms=1secunda 
  
F2:   SETB BUZZ ;APLIC 1 LOGIC LA BUZZ 
 CALL DELAY2 ;ASTEPT 0,25 MILISECUNDE 
 CLR BUZZ ;APLIC 0 LOGIC LA BUZZ 
 CALL DELAY2 ;ASTEPT 0,25 MILISECUNDE 
 DJNZ R2,F2 ;REPET CELE DOUA SEMIPERIOADE (0,25ms + 0,25ms =0,5ms) CE-MI DA O                          
   ;FRECVENTA  
 DJNZ R3,F2 ;A SEMNALULUI AUDIO DE 2 KHz,  DE 2000 ORI , ADICA 
   ;2000*0,5ms=1secunda 
 JMP START ;SARI LA START 
 
DELAY1:   ;RUTINA DE INTARZIERE DE 1,25 MILISECUNDE 

Nivel logic
de tensiune

1 logic

0 logic timp

1perioadaT 2perioadaT

 MOV R4,#3 
 MOV R5,#106 
DELAY10: 
 DJNZ R5,DELAY10 
 DJNZ R4,DELAY10 
 RET 
  
DELAY2:   ;RUTINA DE INTARZIERE DE 0,25 MILISECUNDE 
 MOV R6,#121 
DELAY20: 
 DJNZ R6,DELAY20 
 RET 
  
END 


